Pensieve header: Drawing the Willerton Fish using Gauss diagram formulas from Polyak-Viro.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
Xs[xs_Xs] := xs;
Xs[L_] :=
  Xs @@ PD[L] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]], Xp[l, i], Xm[j, i]];

Xs /: Normalize[xs_Xs] := SortBy[
   xs /. Thread[Union @@ (List @@@ xs) → Range[2 Length[xs]]],
   (List @@ #) &
  ];
Xs /: RotateLeft[xs_Xs] := Normalize[
   xs /. i_Integer :> If[i > 1, i - 1, 2 Length[xs]]
  ];
GPV[F_Xs, K_Xs] /; Length[F] > Length[K] := 0;
GPV[F_Xs, K_Xs] /; Length[F] < Length[K] := Total[
   GPV[F, #] & /@ Subsets[K, {Length[F]}]
  ];
GPV[F_Xs, K_Xs] /; Length[F] == Length[K] := If[
   F === Normalize[K /. Xp | Xm → X],
   (-1)^Count[K, _Xm],
   (*Else*) 0
  ];
GPV[F_Plus, K_] := GPV[#, K] & /@ F;
GPV[c_ * F_Xs, K_] := c GPV[F, K];
V2[K_] := V2[K] = GPV[Xs[X[2, 4], X[3, 1]], Xs[K]];
V3Ds = Expand[Plus[
    1/2 Total[NestList[RotateLeft, Normalize[Xs[X[1, 5], X[4, 2], X[6, 3]]], 5]],
    Total[NestList[RotateLeft, Normalize[Xs[X[1, 4], X[5, 2], X[3, 6]]], 1]]
   ]];
V3[K_] := V3[K] = GPV[V3Ds, Xs@K];
```

```
K = Knot[3, 1]; xs = Xs@K
```

```
KnotTheory::loading : Loading precomputed data in PD4Knots`.
Xs[Xm[4, 1], Xm[6, 3], Xm[2, 5]]
```

```
Normalize /@ Subsets[xs, {2}]
```

```
{Xs[Xm[3, 1], Xm[4, 2]], Xs[Xm[2, 4], Xm[3, 1]], Xs[Xm[1, 3], Xm[4, 2]]}
```

From Polyak-Viro's "Guass Diagram Formulas for Vassiliev Invariants", IMRN 11 (1994) 445-453:

3.B THEOREM 1. *If G is any based Gauss diagram of a knot K then*

(4) $$v_2(K) = \left\langle \text{}, G \right\rangle,$$

where $v_2(K)$ is the Vassiliev invariant of degree 2 which takes values 0 on the unknot and 1 on a trefoil.

```
V2[xs]
```

```
1
```

```
V2 /@ {Knot[3, 1], Mirror@Knot[3, 1]}
```

```
{1, 1}
```

```
V2 /@ AllKnots[{3, 10}]
```

```
{1, -1, 3, 2, -2, -1, 1, 6, 3, 5, 4, 4, 1, -1, -3, 0, -4, -3, -1, -2, 2, 2, -2, 3,
 -1, -3, 1, 0, 4, 1, -1, 1, 5, 2, 0, 10, 4, 9, 7, 6, 7, 5, 0, 8, 8, 4, 1, 7, -1, 2,
 6, -2, 6, -2, 2, 3, -1, 5, 1, 0, 0, 0, 1, 1, -1, 2, -1, 1, -1, 7, 3, -3, 6, 2, -1,
 0, -2, 1, 0, 2, -2, -1, 3, 6, -4, 2, -6, -5, 4, -1, -1, -3, -2, 1, -5, 4, -5, 2,
 3, -4, 2, -2, 1, -3, 1, -4, 3, -2, 0, -3, 2, 3, -4, 1, 2, -1, 0, 3, -4, 1, 3, -1,
 1, 3, -2, 0, 2, 0, -2, 0, 6, 4, 7, -1, 5, 3, 6, 4, 5, 0, 4, -4, -1, -1, -4, 5, 6,
 -3, 4, 7, 0, 2, 2, -3, 1, 2, 1, 0, 0, -2, 4, 3, 5, 6, 3, 0, 1, 2, 2, -1, 0, -1, 1,
 -3, 2, 2, 1, -2, 3, -3, 2, 0, 4, 4, 7, -2, 3, 1, -1, -1, 1, 0, 3, -3, 1, 2, 0, 1,
 1, 0, 2, 0, -1, 6, 1, 2, -2, 8, 3, 5, 1, 7, 2, 4, 0, 3, 1, 6, 3, 0, -2, -3, 9, 2,
 -1, 8, 3, -2, 5, 0, -1, 4, 2, 1, 3, 7, 4, 5, -2, 1, 4, -3, 2, 3, 7, -3, 1, 1, 2}
```

```
Union[Vassiliev[2] /@ AllKnots[{3, 10}] - V2 /@ AllKnots[{3, 10}]]
```

```
{0}
```

From Polyak-
Viro:

4.A THEOREM 2. *If G is a Gauss diagram of a knot K then*

(5) $$v_3(K) = \left\langle \frac{1}{2}\left[\text{}\right] + \left[\text{}\right], G \right\rangle,$$

where $v_3(K)$ is the Vassiliev invariant of degree 3 which takes values 0 on the unknot, $+1$ on the right trefoil and -1 on the left trefoil.

```
V3Ds
```

$$\frac{1}{2}\,\text{Xs}[\text{X}[1,\,4],\,\text{X}[2,\,6],\,\text{X}[5,\,3]] + \text{Xs}[\text{X}[1,\,4],\,\text{X}[3,\,6],\,\text{X}[5,\,2]] + \frac{1}{2}\,\text{Xs}[\text{X}[1,\,5],\,\text{X}[3,\,6],\,\text{X}[4,\,2]] + \frac{1}{2}\,\text{Xs}[\text{X}[1,\,5],\,\text{X}[4,\,2],\,\text{X}[6,\,3]] + \frac{1}{2}\,\text{Xs}[\text{X}[2,\,5],\,\text{X}[3,\,1],\,\text{X}[6,\,4]] + \text{Xs}[\text{X}[2,\,5],\,\text{X}[4,\,1],\,\text{X}[6,\,3]] + \frac{1}{2}\,\text{Xs}[\text{X}[2,\,6],\,\text{X}[4,\,1],\,\text{X}[5,\,3]] + \frac{1}{2}\,\text{Xs}[\text{X}[3,\,1],\,\text{X}[5,\,2],\,\text{X}[6,\,4]]$$

```
{V3[Xs@Knot[3, 1]], V3[Xs@Knot[4, 1]]}
```

```
{-1, 0}
```

```
V3 /@ {Knot[3, 1], Mirror@Knot[3, 1]}
```

```
{-1, 1}
```

```
V3 /@ AllKnots[{3, 7}]
```

```
{-1, 0, -5, -3, 1, 1, 0, -14, -6, 11, 8, -8, -2, -1}
```

```
Vassiliev[3] /@ AllKnots[{3, 7}]
```

```
KnotTheory::loading : Loading precomputed data in Jones4Knots`.
{-1, 0, -5, -3, 1, 1, 0, -14, -6, 11, 8, -8, -2, -1}
```

```
Ks = AllKnots[{3, 11}]; Ks = Ks ⋃ (Mirror /@ Ks);
ListPlot[{V2[#], V3[#]} & /@ Ks, PlotStyle → {Red}, PlotRange → All]
```

KnotTheory::loading : Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory::credits :

The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.



```
ListPlot[
 If[V3[#] ≥ 0, {V2[#], V3[#]}, {V2[Mirror@#], V3[Mirror@#]}] & /@
  AllKnots[{3, 11}],
 PlotStyle → {Red}, PlotRange → All]
```